

# Excel Yourself 2012

---

*By Neale Blackwood*

*[www.a4accounting.com.au](http://www.a4accounting.com.au)*

*[a4@iinet.net.au](mailto:a4@iinet.net.au)*



## Contents

Introduction .....	3
February 2012 - SUBTOTAL.....	4
March 2012 - Sheet Names.....	7
June 2012 - Conditional Formatting.....	10
September 2012 - Print Macro .....	13
November 2012 - TEXT Function .....	14
December 2012 - RANK Function.....	15

## Introduction

This is my second webinar where I'll review some of my past Excel Yourself articles from the INTHEBLACK magazine. This session focusses on the articles from 2012. I won't be covering every single article as time is limited. I have chosen to cover those articles that will benefit most from the webinar format.

Six out of the ten articles will be covered.

Depending on the response and feedback from the webinar I may go back and review previous years' articles as well.

The workings sheets for all ten articles have been included in the companion file for this session. This includes the code for the macro article as well. The macro code can be viewed by pressing Alt + F11 and clicking on the module ThisWorkbook on the left-hand side.

The companion Excel sheets are arranged in date order starting with the February article on the left and working to the right to end with the December article.

In the Excel file the green sheet tabs are the ones to be covered during the session.

I have included some extra content in some of the articles.

The articles were all first published in the CPA Australia INTHEBLACK magazine during 2012.

The content of this manual has been re-written and adapted from the original articles.

## February 2012 - SUBTOTAL

Whilst the SUM function is Excel's most used function it does have some issues when trying to subtotal. The subtotal function is specifically designed to handle subtotalling.

The subtotal function can also perform other calculations apart from SUM and it can also perform calculations on visible cells only. Prior to Excel 2007 it was the only function that could work with visible cells only. Excel 2010 added a new function called AGGREGATE which can also work with visible cells only.

	A	B	C
1	<b>Regions</b>	<b>Values</b>	<b>Values</b>
2	North	20	20
3	South	40	40
5	West	160	160
6	<b>Total</b>	<b>300</b>	<b>220</b>

This figure shows how SUBTOTAL can work with visible cells.

### SUBTOTAL(Function\_Num,Range)

Row 4 is hidden and contains 80 in both columns for East. Cell B6 contains a SUM function whereas cell C6 contains a SUBTOTAL function. To use the SUBTOTAL function on visible cells only, you need to use a different number at the start of the SUBTOTAL function. Typically you use 9, which represents SUM, you need to use 109 when performing a SUM on visible cells only.

The formula in cell C6 is

=SUBTOTAL(109,C2:C5)

This technique is useful when you receive a spreadsheet that has lots of hidden rows and you want to do some checks to make sure that those hidden rows don't contain values that should be made visible.

In terms of subtotalling the SUBTOTAL function has a advantage over the SUM function because it can perform the grand total without having to worry about other totals in the range. The SUBTOTAL function ignores all other SUBTOTAL functions within its range. See the sheet Feb\_SUBTOTAL\_2.

You can add subtotals in manually or there is a built-in feature to add subtotals to tables that have already been sorted. Unfortunately the SUBTOTAL function does not have an icon. It can be quicker to use the AutoSum icon to create a SUM function and then amend the function to use the SUBTOTAL function.

In this example we need a subtotal for state so we can add in manually by inserting a blank row above row 4.

	A	B	C	D
1	<b>City</b>	<b>State</b>	<b>Units</b>	<b>Sales \$</b>
2	Perth	WA	4000	16,000
3	Bunbury	WA	2200	8,000
4	Melbourne	VIC	3500	12,000
5	Geelong	VIC	200	1,000

C
Units
4000
2200
=SUM(C2:C3)
3500

We can use the AutoSum to insert a SUM function with the range and then amend the formula to work with the SUBTOTAL function instead.

Start with the SUM

=SU(C2:C3)
<div> <div> SUBSTITUTE </div> <div> SUBTOTAL </div> <div> SUM </div> </div> <div> 3500 </div> <div> Returns a subtotal in a list or </div>

Delete the M and select the SUBTOTAL

=SUBTOTAL(C2:C3)
<div> <div> SUBTOTAL </div> <div> Returns a subtotal in a list or </div> </div>

=SUBTOTAL(9,C2:C3)
<div> <div> SUBTOTAL </div> <div> Returns a subtotal in a list or </div> </div>

Enter the 9, in front of the range. Done.

Copy across and then copy row 4 to row 7. Then copy row 4 to row 8 and amend the formula in cell C8 to

=SUBTOTAL(9,C2:C7)

3500
200
3700
=SUBTOTAL(9,C6:C7)

1 You can press F2 and edit the formula. You can use the double-headed arrows on the coloured ranges to amend the formula quickly and easily with the mouse.


The colours show you the range involved, but they also allow you to change that range.

## Automated SUBTOTAL

We can remove the subtotals and see how Excel can add them automatically.

If you have a sorted list you can have Excel insert SUBTOTAL functions for you.

Select any cell in the table. Click the Data ribbon tab and click the SUBTOTAL icon on the far left.



The dialog on the following page will display and it will allow you to control the subtotals to be inserted.


The first drop down allows you to set the field that determines where to place the subtotals - when the value in the field changes. This is the field that must be sorted.

The second drop down determines the calculation - usually the SUM.

You then use the check boxes to select the columns to be subtotalled.

Click OK to continue.

The Remove All button will return the list to the layout before the subtotals were added.



Subtotals and grouping buttons are added to the list.

The grouping button on the left allows you to hide and unhide rows quickly.

	A	B	C	D
1	City	State	Units	Sales \$
2	Perth	WA	4000	16,000
3	Bunbury	WA	2200	8,000
4		<b>WA Total</b>	6200	24,000
5	Melbourne	VIC	3500	12,000
6	Geelong	VIC	200	1,000
7		<b>VIC Total</b>	3700	13,000
8		<b>Grand Total</b>	9900	37,000

## March 2012 - Sheet Names

You can't create duplicated sheet names in Excel. This makes it possible to name the sheet after things like department or division. Once you have done that you can have formulas that extract the sheet name. That can then be used to create formula-based reports for department, division or state. The beauty of that is that by copying and renaming the sheet you can create a new report for the new department, division or state.

You can extract the sheet name via a formula. Excel has a CELL function that returns system information.

```
=CELL("filename",A1)
```

It has two arguments. The first is the type of information you want to extract. In our case it is "filename", which returns the full path of the file, including the sheet name. The second argument is a cell reference, A1 in this case.

Cell A2 has the above formula and you can see the result below. I recommend that you always include the reference A1. If you leave it out you may receive incorrect results.

	A
1	
2	E:\A4\Training\Webinars\EY 2012\[Excel Yourself 2012.xlsm]Mar_SheetName

The sheet name always follows the ] symbol, so we can use some text functions to extract the sheet name from the CELL function result.

	A
1	
2	E:\A4\Training\Webinars\EY 2012\[Excel Yourself 2012.xlsm]Mar_SheetName
3	
4	Mar_SheetName

Cell A4 has the following formula

```
=MID(CELL("filename",A1),SEARCH("]",CELL("filename",A1))+1,31)
```

The MID function has three parts to it

**MID(Text,Start\_Num,Num\_Chars)**

**Text** - the text to extract from, usually a cell reference. This formula uses the result of the CELL function as its Text.

**Start\_Num** - the character position number to extract the first character from. In our case it will be the character after the ] symbol. This is supplied by the result of a SEARCH function (to be explained next.)

**Num\_Chars** - how many characters to extract, including the Start\_Num character. 31 is used because that is the maximum length of a sheet name.

The SEARCH function returns the character position of the text being searched for within another text string. It is not case sensitive.

### **SEARCH(Find\_Text,Within\_Text,Start\_Num)**

**Find\_Text** - is usually typed in and is usually a single character. It must be enclosed in quotation marks if typed in. In this case we are looking for the "]" character.

**Within\_Text** - the text to search and it is usually a cell reference. In this case it is the result of the CELL function - the full path of the file, including the sheet name.

**Start\_Num** - optional - only used when multiple entries of Find\_Text exist in the Within\_Text. In our case this is not required.

The SEARCH function returns an error if it can't find the Find\_Text.

The SEARCH function from our formula is

`SEARCH("]",CELL("filename",A1))+1`

This will extract the character position of the ] bracket from the CELL function result and it adds one to the number. We add 1 because we don't want the ] in our sheet name.

So the MID function takes the result of the CELL function and uses the result of the SEARCH function to determine the starting character number of the sheet name. It then extracts 31 characters from that position. In our case there are less than 31 characters but the MID function just extracts the remaining characters.

### **Range Name for Sheet Name**


We can take this function and create a range name that extracts the sheet name. I am not going in to detail about range names as I have an existing webinar dedicated to range names. Range names are a powerful formula feature in Excel.

Copy the formula from cell A4 in the formula bar and press Esc.

Click the Formulas ribbon tab and click the Define Name option.

In the Name box type Sheet\_Name.

Click in the Refers To box





and paste. Press F2, this allows you to edit the formula. Now add a ! in front of both A1 references, as shown below in yellow. Click OK.

```
=MID(CELL("filename",!A1),SEARCH("]",CELL("filename",!A1))+1,31)
```

You can now use the range name Sheet\_Name in any sheet to extract the sheet name.

Using the ! in front of a cell reference is an advanced range name technique. It forces Excel to use the sheet name of the sheet where the range name is used in a formula. This creates a flexible range name that refers to the same cell on different sheets based on where it is used.

### Example application

There is a sheet called Mar\_Data. It has the data for two states, WA and VIC.

There is also a sheet called WA. This extracts data from the Mar\_Data sheet.

Currently cell B1 is an input cell. But we can replace it with the Sheet\_Name formula which then automates the report.

After entering the range name for the sheet name in B1 we can copy the sheet and rename it to VIC and the report will then become the report for VIC.

When creating multiple reports it means you can create one report and simply copy the sheet and rename it and the new report is ready, no other changes required.

## June 2012 - Conditional Formatting

Conditional formats are usually based on the cell that is being formatted. Changing the cell's value may change the format. But you can base the format of one cell on another cell's value. This example is in the Jun\_Cond\_Format sheet.

We have two yellow input cells. Cell C3 for the name and cell C4 asking whether the person is employed.

	A	B	C
1			
2			Input
3		Name	John Smith
4		Employed	Yes
5		Employer Name	


We want the cell colour for C5 to change to yellow if the answer to the question in C4 is yes.

If the answer is anything except yes, it will remain white.

To apply the conditional format click cell C5, then click the Conditional Formatting icon on the Home ribbon.

Click the New Rule option.

Select the last option in the top section to create a formula.





In the Formula box enter the following formula


=C4="yes"

Then click the Format button.

In The Format Cells Dialog click the Fill tab and select the yellow colour and click OK.

**Note:** the check for "yes" is not case sensitive.





Click OK.

Then test the format by changing the entry in cell C4.

The other example from the article involved highlighting entries in a list that didn't appear in another list. This example is on the Jun\_Lists sheet.

	A	B	C	D
1				
2		<b>Invoices_1</b>	<b>Invoices_2</b>	
3		1234	1234	
4		1235	1236	
5		1236	1237	
6		1237	1238	
7		1238	1239	
8		1240	1240	
9		1241	1241	
10		1242	1242	
11				

These two lists are supposed to be the same, but each has an entry that is not in the other.

We can use a formula-based conditional format to check a cell's values against the other list and highlight it if it isn't there.

The function to use in this case is COUNTIF, which performs a conditional count.

### COUNTIF(Range\_To\_Count,Criteria)

**Range\_To\_Count** - is the range that contains the items to count.



**Criteria** - is usually a cell reference that contains the entry to count within the Range\_To\_Count.

In this example the use of relative and fixed (sometimes called absolute) references is important. Relative reference don't have \$ signs whereas a fixed cell reference has two \$ signs.



We need to create two separate conditional formats, one for each list.

1. Select the range B3:B10 click the Conditional Formatting icon on the Home ribbon.

2. Click the New Rule option.
3. Click the last option in the top section "Use a formula to ..."



4. In the Formula box type the following formula, you can use the mouse to select the range and cell references.  
=COUNTIF(\$D\$3:\$D\$10,B3)=0
5. Click the Format button and then click the Fill tab and select the orange colour and click OK.



6. Click OK to apply the format.

Repeat steps 1 to 6 with the range D3:D10 and use the following formula

=COUNTIF(\$B\$3:\$B\$10,D3)=0

These formulas identify if the value isn't in the other range. If the COUNTIF returns zero it means the value isn't in the other list.

Because we selected a range before applying the conditional format we have to use the first cell in that range as the criteria in the COUNTIF functions, and it has to be a relative reference. By using a relative reference each other cell in the range will use its own reference in the COUNTIF formula.


## September 2012 - PrintFooterMacro

It has been on many people's Excel wish list for quite a while, to insert a cell value into the print header or footer. As at Excel 2013, there is still no way to do it automatically, except via a macro.

This macro solution is automatic, so once it is set up it will run every time you print in the file involved. It does assume that macros are enabled.

Macros are a huge topic in Excel and I will only be covering the minimum required to explain how to install and run this macro. Macros are an advanced topic and if you are unfamiliar with them this part of the session may be hard to follow and you may need to review the recording.

Excel can run macros automatically. The way you achieve that is through something called an Event. Excel has a number of Events that it monitors. The Event we can use for our macro solution is a Workbook Event called BeforePrint. This Event runs before you print or use Print Preview.



Press Alt + F11. To see the code, double click the ThisWorkbook item on the left.


```
If ActiveSheet.Name = Sheet5.Name Then
    ActiveSheet.PageSetup.LeftFooter = Sheet5.Range("A1").Value
End If
```

The first line checks to see if the active sheet name is the same as Sheet5. Sheet5 is the system name that Excel uses (it is listed on the left of the above screen shot for the Sep\_PrintMacro sheet).

If the active sheet name is the same as Sheet5 then the second line of code will run, this places the value from cell A1 into the Left Footer. The third line of code ends the If statement.

The entries in the Examples code in the above screen shot show other positions in the header and footer that you can use to place entries.

## November 2012 - TEXT Function



	A	B	C
1	John	Smith	=A1&" "&B1

	A	B	C
1	John	Smith	John Smith

You can join text together in Excel using the & symbol.

This example demonstrates joining together the names from cells A1 and B1 in cell C1 with a space between.

When joining text entries together you don't normally have too many problems. When you start to join numbers and dates together you will most likely experience a few frustrations.

The table below illustrates a few of the frustrations. Column C uses the & symbol to join the entries from columns A and B together as in the above example. Column D uses the TEXT function to achieve the results we are chasing.

	A	B	C	D
1	Text	Number	Joined using &	Joined using TEXT function
2	The profit for the month was	\$156,981	The profit for the month was 156981	The profit for the month was \$156,981
3	Last month's sales increased by	5.17%	Last month's sales increased by 0.0517	Last month's sales increased by 5.17%
4	Report for	November-12	Report for 41214	Report for November-12
5	Forecast sales were	\$2,178,990	Forecast sales were 2178990	Forecast sales were \$2.18 M

## TEXT(Reference,Format)

**Reference** - is usually a cell reference containing a number, or date to format.

**Format** - is enclosed in quotation marks and it uses codes found in the Custom section of the Number tab in the Format Cells dialog.


The formulas in column D above are shown below from top to bottom. The first is in cell D2.

=A2&" "&TEXT(B2,"\$#,###")

=A3&" "&TEXT(B3,"#.##%")

=A4&" "&TEXT(B4,"mmmm-yy")

=A5&" "&TEXT(B5,"\$#.##,")&" M"



## December 2012 - RANK Function

You can use the RANK function to provide a dynamic ranking of values in a range. This rank can then be used as a scoring mechanism. The score is like a golf score, the lower value, the higher your rank, and the better your score.

	A	B	C	D
1	Region	Values	RANK	Correct Ranking
2	NSW	80	1	1
3	VIC	70	2	2
4	WA	60	3	3.5
5	SA	60	3	3.5
6	QLD	50	5	5
7	TAS	40	6	6
8	NT	30	7	7
9	ACT	20	8	8
10	NZ	10	9	9
11				
12		<b>SUM</b>	44	45

The problem comes when there is a tie for a position. The RANK function shares the position between the values that tie.

In this example WA and SA have tied for third place and both have been awarded a 3 as seen in column C. Column C uses the RANK function.

There is no number 4. The next number is 5. What this means is that when adding up results ties allocate lower scores.

When you add up all the values in column C it only adds up to 44. If you add up all the numbers from 1 to 9 it should equal 45.

**Note:** this list is sorted to make it easier to see the ranking, you don't have to sort lists to rank them.

The solution is in column D which averages the results between 3rd and 4th and the score is 3.5 for each rather than 3. This adds up to 45, as it should. Column D uses the RANK.AVG function, a new function included in Excel 2010. This uses the new . (period) in the function name to show it is an amended form of another function.

RANK and RANK.AVG have the same syntax, they just return different results for ties.

**RANK(Reference,Range,Rank\_Type)**

**RANK.AVG(Reference,Range,Rank\_Type)**

**Reference** - usually a relative cell reference. That cell is within the **Range** and is to be ranked against the over values in the **Range**.

**Range** - a range that holds all the values that are to be ranked. Usually a fixed reference, as it shouldn't change as you copy it down.

**Rank\_Type** - optional - this determines if the highest or lowest value in the **Range** is ranked 1. If omitted, or 0 is entered for the **Rank\_Type**, then the highest value is ranked 1. If you the **Rank\_Type** is 1, then the lowest value is ranked 1, like a golf score. It is usually omitted as in our example that follows.

The formula in cell C2 has been copied down and is

`=RANK(B2,$B$2:$B$10)`

This ranks B2 within the range B2:B10.

The formula in cell D2 has been copied down and is

`=RANK.AVG(B2,$B$2:$B$10)`

This also ranks B2 within the range B2:B10, but using the preferred method of averaging the positions rather than duplicating the positions.

In our example there is a tie for 3rd. Rather than showing 3 for both we want to show 3.5 which is the average of positions 3 and 4.

If you need to achieve the same result in Excel 2007 or earlier version you can use the RANK function with the COUNTIF function.

The formula which will work in all versions in D2 is

`=RANK(B2,$B$2:$B$10)+((COUNTIF($B$2:$B$10,B2)-1)*0.5)`

The COUNTIF counts how many times the value appears in the range. It then subtracts 1 because it should appear at least once. It then multiplies the result by 0.5. This in effect, adds 0.5 for each additional tied value.

So if there are two tied entries 0.5 is added to both RANK results.  $(2-1)*0.5 = 0.5$

If there are three tied entries, then 1 is added to each RANK result.  $(3-1)*0.5 = 1$

If there are four tied entries, then 1.5 is added to each RANK result.  $(4-1)*0.5 = 1.5$

The amount added to the duplicated RANK result is the number of tied results, less 1, times 0.5.